

Dowling02.wmxm: Plots & Eqns. in Econ.

TABLE OF CONTENTS

Preface	1
References	2
Isocost Lines	2
Production Possibilities Curves (PPC)	7
Supply and Demand Analysis	13
Income Determination Models	32
IS-LM Analysis	40
Multiple Plots, Plot Sizes	42
Proportional Axes Plots	45
Colors in Plots	47
Local & Global Variables in Maxima Functions	49

```
(%i4) load(draw)$ set_draw_defaults(line_width=2, grid = [2,2], point_type = filled_circle,
    head_type = 'nofilled, head_angle = 20, head_length = 0.5,
    background_color = light_gray, draw_realpart=false)$
fpprintprec : 5$ ratprint : false$
```

1 Preface

Dowling02.wmxm is one of a number of wxMaxima files available in the section "Economic Analysis with Maxima" on my CSULB webpage.

In Dowling02.wmxm, we use Maxima to discuss some simple models in Economics, based mainly on Ch. 2 of the supplemental text: Introduction to Mathematical Economics, 3rd ed, (Schaum's Outline Series), by Edward T. Dowling (1992, 2001), McGraw-Hill.

This modestly priced text is a bargain with many worked out examples. You should compare the examples worked out "by hand" in this text with what we do using Maxima.

We have changed some of the symbols used in particular problems.

An approximate pdf translation (using Microsoft print to pdf) is available as Dowling02fit.pdf. That pdf file can be searched using Ctrl-F.

Ted Woollett
<https://home.csulb.edu/~woollett/>
 woollett@charter.net
 April 21, 2022

2 References

Introduction to Mathematical Economics, 3rd ed, (Schaum's Outline Series),
by Edward T. Dowling (1992, 2001), McGraw-Hill,

Introductory Mathematical Economics, 2nd ed., D. Wade Hands, 2004, Oxford Univ. Press.

Mathematical Economics, 2nd. ed., Jeffrey Baldani, James Bradfield, and Robert W. Turner, 2005, Thomson.

Macroeconomics, 5th ed., Olivier Blanchard, 2011, Prentice Hall.

Fundamental Methods of Mathematical Economics, 4th ed., Alpha C. Chiang and Kevin Wainwright, 2005, McGraw Hill.

(IEA) Introduction to Economic Analysis, R. Preston McAfee, Tracy Lewis, Donald J. Dale, 2009,
read on web: https://saylordotorg.github.io/text_introduction-to-economic-analysis/
get pdf: <https://mcafee.cc/Papers/Introecon/ieav21.pdf>

3 Isocost Lines

An isocost line represents the different combinations of two inputs or factors of production that can be purchased with a given expenditure. The word isocost means "constant cost" or budget.

In a context of purchasing units of capital K and units of labor L , given an available budget E (expenditure), we assume the budget constraint can be expressed in the form:

$$p_k K + p_l L = E,$$

where K and L are respectively units of capital and units of labor, p_k and p_l their respective per unit prices, and E the amount allotted to expenditures (budget). We can assume the prices p_k and p_l are both positive.

In isocost analysis, the individual prices and the expenditure (also positive) are initially held constant; only the different combinations of inputs (K , L) are allowed to change. The constraint can then be graphed by expressing one variable in terms of the other, as seen in the following example and Problems 2.5 and 2.6.

As an example, solve for K in terms of L holding the prices constant:

(%i6) eqn : $p_k K + p_l L = E$

soln : solve (eqn, K);

(soln) $\left[K = -\frac{L p_l - E}{p_k} \right]$

Now express this in the canonical straight line form $y = m*x + b$, with K on the vertical y-axis, L on the horizontal x-axis by expanding the right hand side:

(%i7) `soln, expand;`

(%o7)
$$\left[K = \frac{E}{pk} - \frac{L pl}{pk} \right]$$

When $L = 0$, $K = E/pk$ (all the expenditure goes for capital which is the "y intercept" b in the equation $y = m*x + b$, and the maximum value of K.

The slope (rise over run) is $m = - pl/pk$. As L increases, the number of units of capital K decreases in a straight line (an isocost line). If we compare the effects of two different expenditures E (keeping pl and pk constant), we get two parallel straight lines lines (same slope).

3.1 Prob. 2.5

Prob. 2.5

A person has \$120 to spend on two goods (X, Y) whose respective prices are \$3 and \$5.

(a) Draw a budget line showing all the different combinations of the two goods that can be bought with the given budget (B). What happens to the original budget line (b) if the budget falls by 25 percent, (c) if the price of X doubles, (d) if the price of Y falls to 4?

a.) With $B = 120$, $p_x = 3$, $p_y = 5$, $120 = 3*x + 5*y$. When $x = 0$, $y = y_{max} = 120/5 = 24$. When $y = 0$, $x = x_{max} = 120/3 = 40$.


(%i8) `soln : solve (3*x + 5*y = 120, y);`

(soln)
$$\left[y = -\frac{3x - 120}{5} \right]$$

(%i9) `ys : expand (rhs (soln[1]));`

(ys)
$$24 - \frac{3x}{5}$$

```
(%i10) wxdraw2d (xlabel = "x", ylabel = "y", title = "(a.) B = 120, px = 3, py = 5",
  xrange = [0, 50], yrange = [0, 30], explicit (ys, x, 0, 40))$
```



b.) What happens to this budget line when the budget is reduced by 25%? First find the reduced budget Br, assuming Br is a 25% reduction from 120.

```
(%i11) soln : solve ((120 - Br)/120 = 0.25, Br);
(soln) [Br=90]
```

```
(%i12) Br : at (Br, soln);
(Br) 90
```


We have the same per unit prices, so with the reduced budget we have

$90 = 3x + 5y$. When $y = 0$, $x = x_{\max} = 90/3 = 30$. When $x = 0$, $y = y_{\max} = 90/5 = 18$. Let ysr be the solution for 'y' in case b.

```
(%i13) soln : solve (3*x + 5*y = 90, y);
(soln) [y = - (3x - 90) / 5]
```

```
(%i14) ysr : expand (rhs (soln[1]));
(ysr) 18 - (3x) / 5
```

```
(%i15) wxdraw2d (xlabel = "x", ylabel = "y", title = "(b.) 25% reduction in budget",
  yrange = [0, 30], xrange = [0, 50],
  key = "B = 120, px = 3, py = 5", explicit (ys, x, 0, 40),
  color = red, key = "B = 90, px = 3, py = 5", explicit (ysr,x, 0, 30))$
```



With a reduced budget, less units of both x and y can be purchased, but the slope (rise over run) of the budget line remains the same.

c.) The budget remains $B = 120$, the price of x doubles from 3 to 6, the price of y remains at $py = 5$, so $120 = 6x + 5y$. When $y = 0$, $x = x_{\max} = 120/6 = 20$. When $x = 0$, $y = y_{\max} = 120/5 = 24$. Let the solution for $y(x)$ be called syc .


```
(%i16) soln : solve (120 = 6*x + 5*y, y);
```

```
(soln) [y = - (6x - 120) / 5]
```

```
(%i17) syc : expand (rhs (soln[1]));
```

```
(syc) 24 - (6x) / 5
```

```
(%i18) wxdraw2d (xlabel = "x", ylabel = "y", key = "B = 120, px = 3, py = 5",
title = " (c.) price of x doubles",
yrange = [0, 30], xrange = [0, 50], explicit (ys, x, 0, 40),
color = red, key = "B = 120, px = 6, py = 5", explicit (syc,x, 0, 20),
line_width = 1, color = black, key = "y = 15", explicit (15, x, 0, 15))$
```



(%t18)

Less units of x can be purchased for a given number of units of y purchased.

d.) The price of y falls to 4, with B and px remaining at 120 & 3 respectively, to $120 = 3x + 4y$. When $x = 0$, $y = y_{\max} = 120/4 = 30$. When $y = 0$, $x = x_{\max} = 120/3 = 40$. Let syd be $y(x)$ in case (d).


```
(%i19) soln : solve (120 = 3*x + 4*y, y);
```

```
(soln) [y = - (3 x - 120) / 4]
```

```
(%i20) syd : expand (rhs (soln[1]));
```

```
(syd) 30 - (3 x) / 4
```

```
(%i21) wxdraw2d (xlabel = "x", ylabel = "y", key = "B = 120, px = 3, py = 5",
  title = " (d.) price per unit of y falls to 4 ",
  xrange = [0, 40], xrange = [0, 50], explicit (ys, x, 0, 40),
  color = red, key = "B = 120, px = 3, py = 4", explicit (syd,x, 0, 40), color = black,
  line_width = 1, key = " x = 10", parametric (10, yy, yy, 0, 90/4))$
```



(%t21)

More units of y can be purchased for a given number of units of x.

4 Production Possibilities Curves (PPC)

We adapt the example in Hammock and Wilson, Ch. 2, Sec. 1.

Rather than relate the goods x and y to prices per unit and an overall budget, as in the above section, we consider a nonlinear relation between x and y given by:

$$y = a - b \cdot x^{2.5},$$

where we assume $a > 0$, $b > 0$, so the more units of good x produced, the less units of good y can be produced by a given economy. The constants a and b reflect the production abilities of the economy, including employment of labor and capital and the level of technology available.

Of course we assume both x and y are greater than or equal to zero.

```
(%i24) ymodel : a - b*x^2.5;
  casea : [a = 500, b = 0.01];
  ya : at (ymodel, casea);
```

```
(ymodel) a - b x^2.5
```


```
(casea) [a=500, b=0.01]
```

```
(ya) 500 - 0.01 x^2.5
```

When $y = 0$, $x = x_{\max}$. We can get an approximate value for x_{\max} with a simple plot.

```
(%i26) xm : 100$
wxdraw2d ( explicit (ya, x, 0, xm), color = black,
explicit (0, x, 0, xm))$
```

(%t26)



The value of x_{\max} is between 70 and 80. A purely numerical search for the location at which y_a is zero is provided by Maxima's `find_root (expr, var, v1, v2)` function, in which $v1$ and $v2$ must straddle the location of the root.

```
(%i27) xma : find_root ( ya, x, 70, 80);
```

```
(xma) 75.786
```

```
(%i28) at (ya, x = xma);
```

```
(%o28) 5.6843 10-14
```

An analytic solution to the equation $y_a = 0$, or $0 = a - b \cdot x^{2.5} = a - b \cdot x^{5/2}$, or $x^{5/2} = a/b$, is $x = (a/b)^{2/5}$

```
(%i29) at (a/b, casea);
```

```
(%o29) 5.0 104
```

```
(%i30) %^(2/5);
```

```
(%o30) 75.786
```

When $x = 0$, $x^{2.5} = 0$ and $y = a = 500$.



```
(%i31) 0^2.5;
```

```
(%o31) 0.0
```

A simple plot for casea.

```
(%i33) print (" a = 500, b = 0.01, y = a - b^(2/5)")$
wxdraw2d (xlabel = "x", ylabel = "y", xrange = [0,90], yrange = [0, 600],
explicit (ya, x, 0, xma))$
a = 500, b = 0.01, y = a - b^(2/5)
```

```
(%t33)
```



```
(%i34) xma;
```

```
(%o34) 75.786
```

Define two values of x, (x1 and x2), and a small change in x, (dx).

```
(%i35) [x1, x2, dx] : xma/[5, 2, 20];
```

```
(%o35) [15.157, 37.893, 3.7893]
```

Let $f(x)$ be a Maxima function whose x dependence is the same as that of the Maxima expression ya. We need to place two single quotes (') in front of ya to define $f(x)$.

```
(%i36) f(x) := 'ya;
```

```
(%o36) f(x) := 500 - 0.01 x^2.5
```

We can then define $y1 = f(x1)$, $y1d = f(x1+dx)$, etc. We then display the results in table form using the Maxima matrix function.


```
(%i39) [y1, y1d] : [f(x1), f(x1 + dx)]$
[y2, y2d] : [f(x2), f(x2 + dx)]$
matrix(["1st initial x", "2nd initial x", "change in x"], [x1, x2, dx],
["initial y", "post change y", "change in y"], [y1, y1d, y1d - y1],[y2, y2d, y2d - y2]);
```

1st initial x	2nd initial x	change in x
15.157	37.893	3.7893
initial y	post change y	change in y
491.06	484.38	-6.6807
411.61	387.83	-23.782

We can show these four points on our plot using the points(xL,yL) function

```
(%i42) print (" a = 500, b = 0.01, y = a - b^(2/5)")$
print ("x1 = ", x1, " x2 = ", x2, " dx = ", dx)$
wxdraw2d (xlabel ="x", ylabel = "y", xrange = [0,90], yrange = [0, 600],
explicit (ya, x, 0, xma), color = black, points([x1,x1+dx,x2, x2+dx],[y1,y1d,y2,y2d]))$
a = 500, b = 0.01, y = a - b^(2/5)
x1 = 15.157 x2 = 37.893 dx = 3.7893
```

(%t42)



4.1 Opportunity Cost

More good x can be produced only at the cost of less units of good y and vice versa.

Let oc1 be the opportunity cost in increasing production of good x by dx, starting with production level x1; that is the magnitude of the decrease in units of good y, with a similar definition for oc2.

```
(%i44) [oc1, oc2] := - [y1d - y1, y2d - y2]/dx$
matrix(["Opportunity Cost at x1", "Opportunity cost at x2" ],
       [ oc1, oc2 ]);
(%o44) 
$$\begin{pmatrix} \text{Opportunity Cost at } x1 & \text{Opportunity cost at } x2 \\ 1.7631 & 6.276 \end{pmatrix}$$


```

An opportunity cost function $oca(x)$ for casea can be defined, again using two single quote marks (') in front of $-diff(ya,x)$, which is the opportunity cost per unit increase of production of good x.

```
(%i45) oca(x) := " (- diff (ya, x));
(%o45) oca ( x ) := 0.025 x1.5
(%i46) [oc1, oc2] : [oca(x1), oca(x2)];
(%o46) [1.4753, 5.8315]
```

Here we plot both $oca(x)$ and the points $(x1,oc1)$, $(x2,oc2)$.

```
(%i47) wxdraw2d (xlabel = "initial x", ylabel = "oc(x)", xrange = [0,90], yrange = [0, 20],
                title = "Opportunity Costs: Function and Points",
                explicit (oca, x, 0, xma), color = black, points ([x1,x2], [oc1, oc2]))$
```



The rising curve shows increasing opportunity costs as the initial x increases.

4.2 Shifting PPC

An increase in resources (capital and/or labor) and improved technology can shift the Production Possibilities Curve outward (more units of y for a given number of units of x). Here is an example we call caseb.

If we consider case (b) to have the same value of $a = 500$, but the parameter $b = 0.75 \cdot 10^{-2}$,


```
(%i49) caseb : [a = 500, b = 7.5e-3];
        yb : at (ymodel, caseb);
```

```
(caseb) [a=500,b=0.0075]
```

```
(yb) 500 - 0.0075 x2.5
```

```
(%i50) wxdraw2d (explicit (yb,x,0,100), color = black, explicit (0,x,0,100))$
```

```
(%t50)
```




```
(%i51) xmb : find_root (yb, x, 80,90);
```

```
(xmb) 85.028
```

```
(%i52) fb(x) := "yb;
```

```
(%o52) fb(x) := 500 - 0.0075 x2.5
```

```
(%i54) print ("      y = 500 - b^(2/5)")$
wxdraw2d (xlabel = "x", ylabel = "y", xrange = [0,90], yrange = [0, 600],
         title = " Shifting the PPC Outward with Smaller Parameter b",
         key = "b = 0.01", explicit (ya, x, 0, xma), color = red,
         key = "b = 0.0075", explicit (yb, x, 0, xmb), color = black, line_width = 1,
         key = "", parametric (70, yy, yy, 0, fb(70)))$
      y = 500 - b^(2/5)
```



The magnitude of the slope of the outward shifted PPC (in red) is smaller than that of case a. The vertical line at $x = 70$ shows $y_b(70) > y_a(70)$.

```
(%i55) [f(70), fb(70)];
```

```
(%o55) [90.037, 192.53]
```

5 Supply and Demand Analysis [2.2]

5.1 Demand and Supply Notation: Q_d , Q_s , P_d , P_s

The "market demand curve" can be specified in two equivalent ways:

- 1.) Q_d = quantity (number of units per unit time period) demanded by buyers as a function of the per unit price p ,
- 2.) P_d = the per unit price (\$) buyers are willing and able to pay as a function of the number of units supplied per unit time period q .

For example if $Q_d = a \cdot p^b$, where a and b are parameters and p is price, then $P_d = (q/a)^{1/b}$. P_d is called the "inverse demand curve".

The "market supply curve" can likewise be specified in two equivalent ways:

- 1.) Q_s = the quantity (the number of units per unit time period) producers are willing and able to supply as a function of the per unit price p .
- 2.) P_s = the per unit price at which producers are willing to supply a given quantity of units per unit time period q .

For example if $Q_s = b + c \cdot p$, where b and c are parameters and p is price, then $P_s = (q-b)/c$. P_s is called the "inverse supply curve".

Equilibrium in supply and demand analysis occurs when $Q_s = Q_d$ (or $P_s = P_d$). By equating the supply and demand functions, the equilibrium price and quantity can be determined. See Example 2 and Problems 2.1 to 2.4 and 2.11 to 2.16.

5.2 What is a Demand Curve?

Quoting IEA, Sec. 2.1:

"Eating a French fry makes most people a little bit happier, and we are willing to give up something of value – a small amount of money, a little bit of time – to eat one. What we are willing to give up measures the value – our personal value – of the French fry."

"That value, expressed in dollars, is the willingness to pay for French fries. That is, if you are willing to give up three cents for a single French fry, your willingness to pay is three cents. If you pay a penny for the French fry, you've obtained a net of two cents in value."

"Those two cents – the difference between your willingness to pay and the amount you do pay – is known as consumer surplus. Consumer surplus is the value to a consumer of consumption of a good, minus the price paid."

"..., an order of French fries might be worth \$3 to a consumer, but because French fries are available for around \$1, the consumer obtains a surplus of \$2 in the purchase."


"How much is a second order of French fries worth? For most of us, that first order is worth more than the second one. If a second order is worth \$2, we would still gain from buying it. Eating a third order of fries is worth less still, and at some point we're unable or unwilling to eat any more fries even when they are free, which implies that at some point the value of additional French fries is zero."

"Many, but not all, goods have this feature of diminishing marginal value - the value of the last unit consumed declines as the number consumed increases."

The marginal value function $v(q)$ is the same as what we call the inverse demand function $P_d(q)$.

```
(%i57) qmax : 10$
wxdraw2d (xlabel = "quantity q", ylabel = "v(q)", yrange = [0, 6], key_pos = top_center,
title = " A Demand Curve, v(q) = marginal value of quantity q",
key = "v(q) = (10 - q)/2", explicit ((10 - q)/2, q, 0,qmax), color = red,
line_width = 1, key = " p = per unit price = $3 = v(4)", explicit (3, q, 0, 4),
key = "q0 = 4", color = brown, parametric (4, yy, yy, 0, 3), color = black,
key = "quantity demanded = 4 units ", points ([[4, 3]]),
key = "", label (["p", 0.15, 2.85], ["q0", 3.8, 0.3] ))$
```

(%t57)



Again quoting IEA:

"...given the value p on the vertical axis, the corresponding q_0 on the horizontal axis is the quantity the consumer will purchase."

"It is often important to distinguish the demand curve itself – the entire relationship between price and quantity demanded – from the quantity demanded. Typically, “demand” refers to the entire curve, while “quantity demanded” is a point on the curve."

"Given a price p , a consumer will buy those units with $v(q) > p$, since those units are worth more than they cost. Similarly, a consumer should not buy units for which $v(q) < p$. Thus the quantity q_0 that solves the equation $v(q_0) = p$ gives the quantity of units the consumer will buy."

5.3 What is a Supply Curve?

From IEA Sec. 2.2:

"The supply curve gives the number of units, represented on the horizontal axis, as a function of the price on the vertical axis, which will be supplied for sale to the market."

"...Generally supply is upward sloping, because if it is a good deal for a seller to sell 50 units of a product at a price of \$10 [per unit], then it remains a good deal [for the seller] to supply those same 50 [units] at a price of \$11 [per unit]. The seller might choose to sell more than 50, but if the first 50 weren't worth keeping at a price of \$10, that remains true at \$11.4."

The seller who has a cost $c(q)$ for selling q units obtains a profit π , at price p per unit, of $\pi = p \cdot q - c(q)$, which is just revenue to seller minus cost to seller.

The quantity q which maximizes profit π for the seller is the quantity q satisfying the equation $\partial\pi/\partial q = 0$, or $p - \partial c(q)/\partial q = 0$, or $p = c'(q)$. The first derivative of the cost function, $c'(q)$, is the "marginal cost" to the seller (by definition). The (inverse) supply curve $P_s(q)$ is thus the seller's marginal cost function $c'(q)$. Every point (q,p) on the supply curve returns the maximum profit attainable by the seller at that number of units q . The total cost of producing q units is the sum of the small marginal costs $P_s(q) \cdot dq = c'(q) \cdot dq$ from $q = 0$ to $q = q_{\text{final}}$ which is the definite integral $\int_0^{q_{\text{final}}} (dc(x)/dx, x, 0, q_{\text{final}}) = c(q_{\text{final}}) - c(0) = c(q_{\text{final}})$, since the cost of producing zero units is zero. Thus the cost of producing q_f units is the area under the supply curve from $q = 0$ to q_f .

The revenue from selling q_f units at a price per unit p is $p \cdot q_f$, and the profit from selling q_f units at a price per unit p is profit = revenue - cost = $p \cdot q_f - c(q_f)$. The quantity $p \cdot q_f$ is a rectangle and $c(q_f)$ is the area under the supply curve, so profit is the area inside the rectangle and above the supply curve. The profit is also called the "producer surplus".

Here is an example of an upward sloping nonlinear supply curve given by the equation $P_s(q) = 0.16 \cdot q^2 + 2$. The cost of producing 5 units is the area under the curve from $q = 0$ to $q = 5$.

```
(%i62) Ps : 0.16*q^2 + 2;
      p0 : at (Ps, q = 5);
      Revenue : p0*5;
      Cost : integrate(Ps, q, 0, 5), numer;
      Profit : Revenue - Cost;
```

(Ps) $0.16 q^2 + 2$

(p0) 6.0


(Revenue) 30.0

(Cost) 16.667

(Profit) 13.333

```
(%i64) qmax : 6$
wxdraw2d (xlabel = "q", ylabel = "p", yrange = [0,12], key_pos = top_left,
  title = "A Supply Curve", color = red,
  key = "Ps(q) = 0.16*q^2 + 2", explicit (Ps, q, 0, qmax),
  color = blue, line_width = 1, key = "p = 6", explicit (6, q, 0, 5),
  key = "q0 = 5", color = brown, parametric (5,yy, yy, 0, 6),
  color = black, key = "quantity supplied", points ([ [5, 6] ]),
  label ( ["p", 0.15, 5.75], ["q0", 4.75, 0.45], ["PROFIT = 'Producer Surplus'", 2, 5],
  ["COST = c(q0)", 3, 1.5] ) )$
```

(%t64)



"Anything that increases costs of production will tend to increase marginal cost and thus reduce the supply. For example, as wages rise, the supply of goods and services is reduced, because wages are the input price of labor. Labor accounts for about two-thirds of all input costs, and thus wage increases create supply reductions (a higher price is necessary to provide the same quantity) for most goods and services."

"Costs of materials of course increase the price of goods using those materials. For example, the most important input into the manufacture of gasoline is crude oil, and an increase of \$1 in the price of a 42 gallon barrel of oil increases the price of gasoline about two cents – almost one-for-one by volume."

"Another significant input in many industries is capital, and as we will see, interest is cost of capital. Thus, increases in interest rates increase the cost of production, and thus tend to decrease the supply of goods."

5.4 Example 2: Numerical Linear Demand Curve

An example in which both the demand and supply functions are linear is $Q_d = 10 - 2*p$ and $Q_s = -5 + 3*p$. The solution to $Q_d = Q_s$ is the per unit equilibrium price $p = \$3/\text{unit}$, and equilibrium quantity $q = 4$ units.

```
(%i67) Qs : -5 + 3*p;  
      Qd : 10 - 2*p;  
      soln : solve (Qs = Qd, p);
```

```
(Qs)  3 p - 5  
(Qd)  10 - 2 p  
(soln) [p = 3]
```

```
(%i68) soln : soln[1];  
(soln)  p = 3
```

```
(%i69) qs : at (Qs, soln);  
(qs)  4
```


```
(%i70) ps : at (p, soln);  
(ps)  3
```

For historical reasons, in Economics the per unit price p is placed on the vertical axis and the quantity q is placed on the horizontal axis. To make such a plot using draw2d's explicit function we need the inverse demand function $P_d(q)$ to draw the demand curve and we need the inverse supply function $P_s(q)$ to draw the supply curve.

An alternative is to use the draw2d function implicit. Here is an example of using the implicit function, which has the syntax: `implicit (f(x,y), x, x1, x2, y, y1, y2)`, and which attempts to draw curves such that $f(x,y) = 0$. The three quantities $(x, x1, x2)$ refer to the horizontal axis.

→ `qmax : 10$`
`wxdraw2d (xlabel = "quantity q", ylabel = "price per unit p", yrange = [0, 6],`
`key_pos = top_center, title = " Supply Demand Plot using implicit",`
`key = "Qd = 10 - 2*p", implicit (q - 10 + 2*p, q, 0, qmax, p, 0, 5),`
`color = red, key = "Qs = -5 + 3*p", implicit (q + 5 - 3*p, q, 0, qmax, p, 0, 5),`
`key = "", color = black, line_width = 1, explicit (ps, q, 0, qs), parametric (qs, yy, yy, 0, ps))$`

(%t72)




The alternative to using implicit is to use the inverse demand and supply curves, which for this example are the per unit prices $P_s = (q+5)/3$ and $P_d = (10-q)/2$ as a function of quantity q .

```

→ qmax : 10$
wxdraw2d (xlabel = "quantity q", ylabel = "price per unit p", xrange = [0, 6],
  key_pos = top_center, title = " Supply - Demand Plot using explicit",
  key = "Qd = 10 - 2*p", explicit ((10 - q)/2, q, 0, qmax),
  color = red, key = "Qs = -5 + 3*p", explicit ((q + 5)/3, q, 0, qmax),
  key = "", color = black, line_width = 1, explicit (ps, q, 0, qs), parametric (qs, yy, yy, 0, ps ))$

```



5.5 Numerical Nonlinear Demand Curve Example

We consider an example from Hammock & Mixon, Ch.2, in which $Q_d = 1200/p^{5/2}$ and $Q_s = -5 + 10*p$. For p close to zero, Q_d becomes arbitrarily large, so we want to stay away from the point $p = 0$. This model of Q_d is unrealistic for small p .

5.5.1 Graphical Exploration

We first map Q_d onto a list of prices. To do this, one way is to first define fQ_d as a formal Maxima function:

Q_d is here defined as a Maxima expression depending on the parameter p .

```
(%i75) Qd : 1200/p^(5/2);
```

```
(Qd) 
$$\frac{1200}{p^{5/2}}$$

```

Placing two single quotes ('') in front of Q_d allows the following method of defining a Maxima function $fQ_d(p)$.

```
(%i76) fQd(p) := "Qd";
```

```
(%o76) fQd(p) :=  $\frac{1200}{p^{5/2}}$ 
```

Let pL be a list of prices.

```
(%i77) pL : makelist(j, j, 1, 20);
```


```
(pL) [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20]
```

```
(%i78) QdL : map('fQd, pL), numer;
```

```
(QdL) [1200.0,212.13,76.98,37.5,21.466,13.608,9.2563,6.6291,4.9383,
3.7947,2.9902,2.4056,1.9693,1.6363,1.3771,1.1719,1.0071,0.87297,
0.7626,0.67082]
```

```
(%i79) wxdraw2d (xlabel = "p", ylabel = "Qd", xrange = [0,30], yrange = [0, 100], points(pL, QdL))$
```

```
(%t79)
```



```
(%i80) Qs : -5 + 10*p;
```

```
(Qs) 10 p - 5
```


```
(%i81) fQs(p) := "Qs";
```

```
(%o81) fQs(p) := 10 p - 5
```

```
(%i82) QsL : map('fQs, pL);
```


```
(QsL) [5,15,25,35,45,55,65,75,85,95,105,115,125,135,145,155,165,
175,185,195]
```

```
(%i83) wxdraw2d (xlabel = "p", ylabel = "Qs", xrange = [0, 30], yrange = [0, 200],
color = red, points(pL, QsL))$
```




We first make a crude plot to locate the region where $Q_d = Q_s$.

```
(%i84) wxdraw2d (xlabel = "p", ylabel = "q", yrange = [0,300], key = "Qd", points(pL, QdL),
color = red, key = "Qs", points (pL, QsL))$
```



We see that $Q_d = Q_s$ at about $p = 4$. Now that we know the p region in which to look, we make a plot over a small region of p .

```
(%i85) wxdraw2d(xlabel = "p", ylabel = "q", yrange = [20, 70], key = "Qd",
  explicit (Qd, p, 3, 5), color = red, key = "Qs", explicit (Qs, p, 3, 5))$
```



We need to use `find_root (f(x), x, x1, x2)` to locate the root of the nonlinear equation $Q_d - Q_s = 0$.

```
(%i86) ps : find_root (Qd - Qs, p, 4, 4.3);
```

```
(ps) 4.0765
```

```
(%i87) qs : at (Qs, p = ps);
```

```
(qs) 35.765
```

So the equilibrium solution implies a per unit price $p \sim 4.1$ and $q \sim 35.8$ units per unit time period.

5.5.2 Plot Using Inverse Demand and Supply Curves

With $Q_d = 1200/p^{5/2}$, we get $P_d = (1200/q)^{2/5}$. With $Q_s = -5 + 10 \cdot p$, we get $P_s = (q+5)/10$. We make a plot with prices (\$ per unit) on the vertical axis and Quantity (units per unit time period) on the horizontal axis.

```
(%i88) Pd : (1200/q)^(2/5);
```

```
(Pd) 
$$\frac{1200^{2/5}}{q^{2/5}}$$


```

```
(%i89) at (Pd, q = qs), numer;
```

```
(%o89) 4.0765
```



```
(%i90) wxdraw2d (xlabel = "Quantity: units per unit time period", ylabel = "$ per unit",
  yrange = [3.5, 4.6],
  key = "Pd", explicit ( Pd, q, 34, 37), color = red, key = "Ps",
  explicit ( (q + 5)/10, q, 34, 37 ),key = "", color = black, line_width = 1,
  explicit (ps, q, 34, qs), parametric (qs,yy,yy, 3.5, ps))$
```



With the equilibrium quantity $q_s \sim 35.76$ units per unit time period, for $q > q_s$ the supply curve Q_s lies *above* the demand curve Q_d (in a plot with price on the vertical axis and quantity on the horizontal axis). The price per unit that producers would have to be paid in order to produce a quantity $q > q_s$ exceeds the price per unit that buyers are willing and able to pay for that quantity.

For $q < q_s$, the demand curve lies above the supply curve.

5.6 Prob. 2.1, 2.2, 2.3

Prob 2.1.

A complete demand function is given by the equation

$$Q_d = -30p + 0.05Y + 2pr + 4T$$

where p is the per unit price of the good, Y is income, T is taste, and pr is the price of a related good. The variable pr appears with a positive coefficient, which means if the price of the related good increases, the quantity of the original good demanded increases.

This means the original good and the related good are "substitute goods" (think of Pepsi and Diet Coke).

If the coefficient of pr is a negative number, a rise in price of the related good results in a decrease in demand of the original good. This is an example of a "complementary good" (think rice and beans) since demand for both goods rise and fall together.

Q_d is the dependent variable. The other variables are independent variables. Can this function be graphed?

Since the complete function contains four different independent variables (p , Y , pr , T), it cannot be graphed as is. In ordinary demand analysis, however, it is assumed that all the independent variables except price are held constant, so that the effect of a change in price on the quantity demanded can be measured under constant Y , pr , and T conditions. If the variables Y , pr , and T are held constant, then we can plot Q_d vs. p , or P_d vs. q .

Prob 2.2 + 2.3

(a) Draw the graph for the demand function in Problem 2.1, assuming $Y = 5000$, $pr = 25$, and $T = 30$.

(b) What does the typical demand function drawn in part (a) show?

(c) What happens to the graph if the price per unit of the good changes from 5 to 6?

(d) What happens if any of the other variables change? For example, if income increases to 7400?

a) We first define the demand function model as the expression Q_s_gen .

```
(%i91) Qd_gen : -30*p + 0.05*Y + 2*pr + 4*T;
```

```
(Qd_gen) 2 pr - 30 p + 0.05 Y + 4 T
```

The parameter values for case a are contained in the list $case_a$.

```
(%i92) case_a : [ Y = 5000, pr = 25, T = 30 ];
```

```
(case_a) [ Y=5000, pr=25, T=30 ]
```

Let Q_d_a be the demand function expression for case a.

```
(%i93) Qd_a : at (Qd_gen, case_a);
```

```
(Qd_a) 420.0 - 30 p
```

When $Q = 0$, $p = 420/30 = 14$, the maximum price per unit. When $p = 0$, $Q = 420$, the maximum quantity.

```
(%i94) 420/30;
```

```
(%o94) 14
```

```
(%i95) psoln : solve (q = Qd_a, p);
```

```
(psoln) [p = -  $\frac{q-420}{30}$ ]
```

Let Pd_a be the inverse demand curve depending on quantity q .

```
(%i96) Pd_a : expand (rhs (psoln[1]));
```

```
(Pd_a) 14 -  $\frac{q}{30}$ 
```


```
(%i98) qmax : 500$
```

```
wxdraw2d ( xlabel = "q", ylabel = "p", yrange = [0, 16],
```

```
title = "Case a: Y = 5000, pr = 25, T = 30",
```

```
key = " inverse demand function Pd ", explicit (Pd_a, q, 0, qmax))$
```

```
(%t98)
```



b) What does the typical demand function drawn in part (a) show?

The demand function graphed in part (a) shows all the different quantities of the good that will be demanded at different prices, assuming a given level of income, taste, and price of substitutes (here 5000, 30, 25) which are not allowed to change.

c) What happens to the graph if the price per unit of the good changes from 5 to 6?

If nothing changes but the price of the good, the graph remains exactly the same since the graph indicates the different quantities that will be demanded at all the possible prices. A simple change in the price of the good occasions a movement along the curve which is called a "change in quantity demanded." When the price goes from 5 to 6, the quantity demanded falls from 270 to 240, a movement from A to B on the curve.

$A = (Q = 270, p = 5) \rightarrow B = (Q = 240, p = 6)$.

```
(%i99) q5 : at(Qd_a, p = 5);
```

```
(q5) 270.0
```


```
(%i100) q6 : at (Qd_a, p = 6);
```

```
(q6) 240.0
```

```
(%i102) qmax : 500$
```

```
wxdraw2d ( xlabel = "q", ylabel = "p", yrange = [0, 16],
  title = "Case a: Y = 5000, pr = 25, T = 30",
  key = " inverse demand function Pd ", explicit (Pd_a, q, 0, qmax), color = black,
  key = "", points([ [270, 5], [240, 6] ]), label ( ["A", q5, 6] ), label ( ["B", q6, 7] ),
  line_width = 1, explicit (5, q, 0, q5), parametric (q5, yy, yy, 0, 5), color = red,
  explicit (6, q, 0, q6), parametric (q6, yy, yy, 0, 6))$
```

```
(%t102)
```



(d) What happens if income increases to 7400?

If any of the other variables change, there will be a shift in the curve. This is called a "change in demand" because it results in a totally new demand function (and curve) in response to the changed conditions. If income increases to 7400, the new demand function becomes $Qd = -30 \cdot p + 540$, or $Pd = -(q - 540)/30$. This is graphed as a red line.

Call this case b.

```
(%i103) case_b : subst (7400, 5000, case_a);
(case_b) [Y=7400,pr=25,T=30]
```

```
(%i104) Qd_b : at (Qd_gen, case_b);
(Qd_b) 540.0 - 30 p
```

When $p = 0$, $q = 540$.

```
(%i105) psoln : solve (q = Qd_b, p);
(psoln) [p = -\frac{q-540}{30}]
```

Let Pd_b be the inverse demand curve, depending on quantity q , for case b.

```
(%i106) Pd_b : expand (rhs (psoln[1]));
(Pd_b) 18 - \frac{q}{30}
```

When $q = 0$, $p = 18$.


If we compare the quantity demanded Qd in cases a and b for the price per unit fixed at $p = 10$. Let qa and qb denote these quantities.

```
(%i107) qa : at (Qd_a, p = 10);
(qa) 120.0
```

```
(%i108) qb : at (Qd_b, p = 10);
(qb) 240.0
```

The quantity demanded at $p = 10$ is doubled.

```
(%i110) qmax : 550$
wxdraw2d ( xlabel = "q", ylabel = "p", yrange = [0, 20],
  key = "Case a: Y = 5000, pr = 25, T = 30", explicit (Pd_a, q, 0, qmax),
  color = red, key = "Case b: Y = 7400, pr = 25, T = 30", explicit (Pd_b, q, 0, qmax),
  color = black,
  key = "", points([ [270, 5], [240, 6] ]), label ( ["A", q5, 6] ), label ( ["B", q6, 7] ),
  line_width = 1, explicit (5, q, 0, q5), parametric (q5, yy, yy, 0, 5), color = brown,
  explicit (10, q, 0, qb))$
```



With an increase in income Y the demand curve shift outward but maintains the same slope.

5.7 Symbolic Linear Demand and Supply Analysis

We follow BBT, p. 3.

We assume a model in which the (inverse) demand and supply equations have the linear forms:

$$P_d = a - b \cdot q, \quad P_s = c + d \cdot q,$$

in which a , b , c , and d are all greater than zero and $a > c$ so that the equilibrium price will be given by demand = supply, or $P_d = P_s$, which implies $q_s = (a - c)/(b + d) > 0$, and $p_s = (a \cdot d + b \cdot c)/(b + d)$.

```
(%i113) Pd : a - b*q;
Ps : c + d*q;
soln : solve (Pd = Ps, q);
```

(Pd) $a - b \cdot q$

(Ps) $d \cdot q + c$

(soln) $\left[q = -\frac{c - a}{d + b} \right]$

(%i114) `qs : rhs (soln[1]);`

$$(qs) \quad - \frac{c-a}{d+b}$$

(%i115) `num(qs)/denom(qs);`

$$(%o115) \quad \frac{a-c}{d+b}$$

Equilibrium quantity.

(%i116) `qs : %;`

$$(qs) \quad \frac{a-c}{d+b}$$

(%i117) `at (Pd, q = qs);`

$$(%o117) \quad a - \frac{b(a-c)}{d+b}$$

(%i118) `ratsimp(%);`

$$(%o118) \quad \frac{a d + b c}{d+b}$$

Equilibrium price per unit.

(%i119) `ps : %;`

$$(ps) \quad \frac{a d + b c}{d+b}$$

We calculate the rate of change of qs and ps with a , $\partial(qs)/\partial a$ and $\partial(ps)/\partial a$, holding b , c , and d constant.

(%i120) `[diff(qs,a), diff(ps,a)];`

$$(%o120) \quad \left[\frac{1}{d+b}, \frac{d}{d+b} \right]$$

The parameter a is the "demand intercept", the price when $q = 0$ (plotting p on the vertical axis and q on the horizontal axis). We then see that an increase in the value of the demand intercept leads to an increase in both quantity and price (in general).

We next look at the effect of an increase in the parameter d , the slope of the supply curve, on qs and ps by looking at the partial derivatives $\partial(qs)/\partial d$ and $\partial(ps)/\partial d$, holding a , b , and c constant.

(%i121)[diff (qs, d), diff (ps, d)], ratsimp;

(%o121) $\left[\frac{c-a}{d^2+2bd+b^2}, -\frac{bc-ab}{d^2+2bd+b^2} \right]$

Since $a > c$, $\partial(qs)/\partial d < 0$ and $\partial(ps)/\partial d > 0$ in general. The change in the equilibrium price ps is in the same direction as the change in the parameter d and the change in the equilibrium quantity qs is opposite in sign to the direction of change in d . Thus a decrease in the supply slope d will lead to an increase in the equilibrium quantity qs but to a decline in equilibrium price ps .

Quoting Baldani, Bradfield, & Turner, p. 6:

"A decrease in the supply curve slope d might result from such factors as improved production technology, lowered input prices, or entry of new firms into the industry, all of which increase the the quantity supplied."

6 *Income Determination Models [2.3]*

We consider the market for goods in which consumers, investors, and the government all compete for goods. Equilibrium in that market for goods requires that the quantity of goods supplied (GNP), which is the national income Y , be equal to the aggregate demand for goods, which, (neglecting exports and imports) is $C + I + G$ (a three-sector economy).

Income determination models which include export and import express the equilibrium level of (national) income Y in a four-sector economy as (each component measured in dollars):

$$Y = C + I + G + (X - Z)$$

where Y = national income (GDP), C = (personal) consumption of goods and services, I = investment in the means for future production (such as purchase of upgraded technology), G = government expenditures on goods and services (but not social security or welfare), X = exports of nationally produced goods and services, and Z = imports of goods and services produced in other countries.

By substituting the information supplied in the problem, it is an easy matter to solve for the equilibrium level of income Y . Aggregating (summing) the variables on the right allows the equation of be graphed in a two-dimensional space.

See Example 3 and Problems 2.7 to 2.10 and 2.17 to 2.22.

6.1 Example 3

 Example 3 considers a "two sector economy" in which only consumption (C) and investment (I) in the means of future production are taken into account.

We assume

$$Y = C + I,$$

with $C = C_0 + b \cdot Y$, where b is the "marginal propensity to consume" (MPC) and $I = I_0 =$ constant independent of Y .

Solving for Y gives

$$Y = (C_0 + I_0) / (1 - b) = (C_0 + I_0) / (1 - \text{MPC}).$$

The marginal propensity to consume (MPC) is the proportion of an aggregate raise in pay that a consumer spends on the consumption of goods and services, as opposed to saving (and hence investing) it.

In general the definition of MPC is:

$$\text{MPC} = dC/dY = (\text{first derivative of consumption with respect to } Y).$$

C is called the "consumption function" and the sum $C + I$ is the "aggregate demand function" D .

MPC is the key determinant of the "Keynesian multiplier", which describes the effect of increased investment or government spending as an economic stimulus.

A numerical example is: $C_0 = 85$, $b = 0.9$, and $I_0 = 55$. Then $C_0 + I_0 = 140$, $1/(1-b) = 10$, $Y = 1400$, $C = 85 + 0.9 \cdot 1400 = 1260$, and $C/Y = 0.9$.

C_0 and I_0 are "autonomous expenditures", since they are constant and are not affected by the size of the national income Y . The factor $1/(1-b)$ is the "autonomous expenditure multiplier" which is a measure of the multiple effect of each dollar of autonomous spending on the equilibrium level of national income. Y is an "endogenous variable", and C_0 , I_0 , and b are "exogenous variables".


6.2 Prob. 2.7

Assume $Y = C + I$, with $C = 50 + 0.8 \cdot Y =$ "consumption function", and $I = I_0 = 50$.

(a) Plot the consumption function

```
(%i122) wxdraw2d ( key_pos = top_left, key = "C = 50 + 0.8*Y ", yrange = [0, 600],
explicit (50 + 0.8*Y, Y, 0, 600), ylabel = "C", xlabel = "Y", grid = [2,2] );
```

(%t122)




(%o122)

(b) plot the "aggregate demand function" $C + I_0$.

Ans: $C + I_0 = 100 + 0.8 Y$ is a line parallel (same slope) to the consumption function

```
(%i123) wxdraw2d ( xlabel = "Y", yrange = [0, 600], key_pos = top_left,
key = "C = 50 + 0.8*Y ", explicit (50 + 0.8*Y, Y, 0, 600), color = red,
key = "C + I_0, I_0 = 50", explicit (100 + 0.8*Y, Y, 0, 600) );
```

(%t123)




(%o123)

(c) A graphical method to find the equilibrium value of Y is to use equal units for Y on the horizontal axis and aggregate demand $C + I_0$ on the vertical axis, and then draw a 45 deg line from the origin and locate its intersection with the aggregate demand ($C + I_0$) line. Thus find $Y = 500$.

We use the `proportional_axes = xy` option for such a "graphical solution".

```
→ wxdraw2d ( xlabel = "Y", yrange = [0, 600], key_pos = top_left, proportional_axes = xy,
            key = "C = 50 + 0.8*Y", explicit (50 + 0.8*Y, Y, 0, 600), color = red,
            key = "C + I0, I0 = 50", explicit (100 + 0.8*Y, Y, 0, 600), key = "", color = black,
            line_type = dashes, explicit (Y, Y, 0, 600), point_size = 1.5, points ([[500, 500]]),
            line_width = 1, explicit (500, Y, 0, 500), parametric (500, t, t, 0, 500) ),
            wxplot_size = [480,480];
```



(%t124)

(%o124)

In comparison with the graphical solution (some work) is the painless algebraic solution:

```
(%i125) solve (Y = 100 + 0.8*Y);
```

```
(%o125) [Y=500]
```

6.3 Prob. 2.9 Lump Sum Tax Effect


Consider the effects of a lump sum tax T (a tax independent of income) on national income Y in a two sector economy $Y = C + I$, with $C = C_0 + b*Y_d$, $Y_d = \text{disposable income} = Y - T$.

Without the tax, $Y_d = Y$, $Y = C_0 + b*Y + I_0$, $Y = (C_0+I_0)/(1 - b) = D_0/(1-b)$.

With the tax $Y_d = Y - T$, $Y = C_0 + b*(Y - T) + I_0 = D_0 - b*T + b*Y$, $Y_t = (D_0 - B*T)/(1-b) < Y$.

For a numerical example, with $C_0 = 100$, $b = 0.6$, $I_0 = 40$, $T = 50$, $1-b = 0.4$, $D_0 = 140$, $b*T = 30$. Without tax $Y = 140/0.4 = 350$. With tax $Y = (140 - 30)/0.4 = 110/0.4 = 275$. The effect of the lump sum tax $T = 50$ causes Y to decrease from 350 to 275.

```
(%i126) wxdraw2d ( xlabel = "Y", xrange = [0, 600], yrange = [0, 600],
  key_pos = top_left, proportional_axes = xy,
  key = "C + I = 140 + 0.6*Y", explicit (140 + 0.6*Y, Y, 0, 600), color = red,
  key = "C + I = 110 + 0.6*Y", explicit (110 + 0.6*Y, Y, 0, 600), key = "", color = black,
  line_type = dashes, explicit (Y, Y, 0, 600), point_size = 1.5, points ( [275, 350], [275, 350] ),
  line_width = 1, explicit (275, Y, 0, 275), parametric (275, yy, yy, 0, 275),
  explicit (350, Y, 0, 350), parametric (350, yy, yy, 0, 350) ), wxplot_size = [480,480];
```



(%o126)

6.4 Prob. 2.10: Effects on Y of a Proportional + Lump Sum Tax

Explain with the aid of a graph how the incorporation of a proportional tax (a tax depending on income) influences the parameters of the income determination model.

1. Without the tax, $Y = C + I$, $C = 85 + 0.75*Y$, $I_0 = 30$, Aggregate demand $D = C + I$, $D = 115 + 0.75*Y$, $Y = 115/(1 - 0.75) = 460$.

```
(%i127) solve (Y = 85 + 0.75*Y + 30);
```

```
(%o127) [Y=460]
```

2. With the tax, $C = 85 + 0.75*Y_d$, $Y_d = Y - T$, $T = 20 + 0.2*Y$, $Y_d = Y - 20 - 0.2*Y = 0.8*Y - 20$,
 $C = 85 + 0.75*(0.8*Y - 20) = 70 + 0.6*Y$, $Y = C + I_0 = 100 + 0.6*Y$, $Y = 100/(1 - 0.6) = 250$.

```
(%i132) T : 20 + 0.2*Y;
      Yd : Y - T;
      I0 : 30;
      C : 85 + 0.75*Yd;
      solve (Y = C + I0);
```

```
(T) 0.2 Y + 20
```

```
(Yd) 0.8 Y - 20
```


```
(I0) 30
```

```
(C) 0.75 (0.8 Y - 20) + 85
```

```
(%o132) [Y=250]
```

```
→ wxdraw2d ( xlabel = "Y", xrange = [0, 600], yrange = [0, 600],
      key_pos = top_left, proportional_axes = xy,
      key = "C + I = 115 + 0.75*Y", explicit (115 + 0.75*Y, Y, 0, 600), color = red,
      key = "C + I = 100 + 0.6*Y", explicit (100 + 0.6*Y, Y, 0, 600), key = "", color = black,
      line_type = dashes, explicit (Y, Y, 0, 600), point_size = 1.5,
      points ( [250, 460], [250, 460] ), line_width = 1, explicit (250, Y, 0, 250),
      parametric (250, yy, yy, 0, 250), explicit (460, Y, 0, 460),
      parametric (460, yy, yy, 0, 460) ), wxplot_size = [480,480];
```

```
(%t133)
```



```
(%o133)
```

Incorporation of the proportional tax lowers the slope (MPC) of the C + I line from 0.75 to 0.6. The vertical intercept is also lowered because the tax structure includes a lump-sum tax of 20. Because of the tax structure, the equilibrium level of national income falls from 460 to 250.

6.5 Prob. 2.21

(a.) Find the reduced form for the equilibrium income in terms of C_0 , I_0 , T_0 , b , and t if $Y = C + I$, $I = I_0$, $C = C_0 + b \cdot Y_d$, $Y_d = Y - T$, $T = T_0 + t \cdot Y$.

In Maxima, define these quantities in reverse order:

```
(%i138) kill(I0)$
      T : T0 + t*Y;
      Yd : Y - T;
      C : C0 + b*Yd;
      soln : solve (Y = C + I0, Y);
```

```
(T)   Y t + T0
```

```
(Yd)  - Y t + Y - T0
```

```
(C)   b (- Y t + Y - T0) + C0
```

```
(soln) [Y = -  $\frac{T_0 b - I_0 - C_0}{b t - b + 1}$ ]
```

```
(%i139) Ye : rhs (soln[1]);
```

```
(Ye)  -  $\frac{T_0 b - I_0 - C_0}{b t - b + 1}$ 
```

```
(%i140) num(Ye)/denom(Ye);
```

```
(%o140)  $\frac{-T_0 b + I_0 + C_0}{b t - b + 1}$ 
```

```
(%i141) Ye : %;
```

```
(Ye)   $\frac{-T_0 b + I_0 + C_0}{b t - b + 1}$ 
```

(b.) Evaluate Y_e for $I_0 = 30$, $C_0 = 85$, $b = 0.75$, $t = 0.2$, $T_0 = 20$.

```
(%i142) at (Ye, [I0 = 30, C0 = 85, b = 0.75, t = 0.2, T0 = 20]);
```

```
(%o142) 250.0
```

(c.) The multiplier is changed from $1/(1 - b)$ to $1/(1 - b + b \cdot t)$. This reduces the size of the multiplier because it makes the denominator larger and the fraction smaller.

(%i143) at (1/(1-b), b = 0.75);

(%o143) 4.0

(%i144) at (1/(1 - b + b*t), [b = 0.75, t = 0.2]);

(%o144) 2.5

6.6 Prob 2.22 Income Determination Model with a Foreign Sector

Assume $Y = C + I + G + (X - Z)$, with $C = C_0 + b*Y$, $Z = Z_0 + z*Y = \text{imports}$, $I = I_0$, $G = G_0$, $X = \text{exports} = X_0$.

(a.) Find the reduced symbolic form for Y in terms of a multiplier.

(%i147) $Z : Z_0 + z*Y;$

$C : C_0 + b*Y;$

soln : solve ($Y = C + I_0 + G_0 + X_0 - Z$, Y);

(Z) $Y z + Z_0$

(C) $Y b + C_0$

(soln) $[Y = -\frac{Z_0 - X_0 - I_0 - G_0 - C_0}{z - b + 1}]$

(%i148) $Ye : \text{rhs}(\text{soln}[1]);$

(Ye) $-\frac{Z_0 - X_0 - I_0 - G_0 - C_0}{z - b + 1}$

(%i149) $\text{num}(Ye)/\text{denom}(Ye);$

(%o149) $\frac{-Z_0 + X_0 + I_0 + G_0 + C_0}{z - b + 1}$

(%i150) $Ye : %;$

(Ye) $\frac{-Z_0 + X_0 + I_0 + G_0 + C_0}{z - b + 1}$

(b.) Consider the following numerical inputs and find Ye .

(%i152) case : [I0 = 90, G0 = 65, X0 = 80, C0 = 70, Z0 = 40, b = 0.9, z = 0.15]\$
at (Ye, case);

(%o152) 1060.0

(c.) Compare the multiplier without and with the inclusion of an import term proportional to Y .

(%i154) at(1/(1-b), case);
 at (1/(1 - b + z), case);
 (%o153) 10.0
 (%o154) 4.0

Introduction of the "marginal propensity to import" z into the model reduces the size of the multiplier, in this example from 10 to 4.

7 *IS-LM Analysis [2.4]*

The "IS schedule" is the locus of points representing all the different combinations of interest rates and income levels consistent with equilibrium in the goods (commodity) market. Equilibrium in the goods market demands that planned "investment" and planned "saving" be equal, which is where "IS" comes from.

The "LM schedule" is the locus of points representing all the different combinations of interest rates and income levels consistent with equilibrium in the money market. The "L" stands for "liquidity preference" (as the demand for money is sometimes known) and the "M" stands for the quantity of money supplied.

IS-LM analysis seeks to find the level of income and the rate of interest at which both the commodity market and the money market will be in equilibrium simultaneously. This can be accomplished with the techniques used for solving simultaneous equations.

Unlike the simple income determination model in Section 2.3, IS-LM analysis deals explicitly with the interest rate (i) and incorporates its effect into the model. See Example 4 and Problems 2.23 and 2.24.

The commodity market for a simple two-sector economy is in equilibrium when $Y = C + I$. The money market is in equilibrium when the supply of money (M_s) equals the demand for money (M_d).

The demand for money (M_d) is composed of the transaction-precautionary demand for money (M_t) and the speculative demand for money (M_z).

$$M_d = M_t + M_z.$$

7.1 Example 4, Numerical Example

Assume a two-sector economy where national income $Y = C + I$, personal consumption $C = 48 + 0.8Y$, investment $I = 98 - 75i$, the supply of money is $M_s = 250$, the transaction-precautionary demand for money is $M_t = 0.3Y$, and the speculative demand for money is $M_z = 52 - 150i$. We use i for the (decimal) interest rate on money, following Dowling's notation.

Commodity equilibrium (IS) exists when $Y = C + I$, or $Y - C - I = 0$.

Monetary equilibrium (LM) exists when $M_s = M_d = M_t + M_z$, or $M_s - M_t - M_z = 0$.

We can solve these two equations for Y and interest rate i .

```
(%i163) kill(Y,i)$
C : 48 + 0.8*Y;
I : 98 - 75*i;
IS : Y - C - I;
Ms : 250;
Mt : 0.3*Y;
Mz : 52 - 150*i;
LM : Ms - Mt - Mz;
soln : solve ([IS, LM]);

(C) 0.8 Y + 48
(I) 98 - 75 i
(IS) 75 i + 0.2 Y - 146
(Ms) 250
(Mt) 0.3 Y
(Mz) 52 - 150 i
(LM) 150 i - 0.3 Y + 198
(soln) [[i = 2/25, Y = 700]]
```

```
(%i164) soln : soln[1];
(soln) [i = 2/25, Y = 700]
```

```
(%i165) float (soln);
(%o165) [i = 0.08, Y = 700.0]
```

Check solution.

```
(%i166) at ([C, I, C+I, Mt, Mz], soln), numer;
(%o166) [608.0, 92.0, 700.0, 210.0, 40.0]
```

The commodity and money markets will thus be simultaneously in equilibrium when $Y = 700$ and $i = 0.08$.

7.2 Solution for a Symbolic IS-LM Model

Assume four exogenous variables are three levels of autonomous spending C_0 , I_0 , and G_0 , and the constant stock of money M_s . The four endogenous variables to be determined by the model are Y , C , I , and r (decimal rate of interest).

The four equations (the first three equations clear the goods market, and the fourth equation clears the money market):

$$Y = C + I + G_0,$$

$$C = C_0 + a Y,$$

$$I = I_0 + e r,$$

$$M_s = f Y - g r,$$

imply two simultaneous equations for unknowns (Y , r).

The solution for Y and r will depend on C_0 , I_0 , G_0 , and will be expressed in terms of the parameters f , g , e , and a .

```
(%i172) kill(Y, r)$
C : C0 + a*Y;
I : I0 + r*e;
eq1 : Y = C + I + G0;
eq2 : Ms = f*Y - r*g;
soln : solve ([eq1, eq2],[Y, r]);

(C)   Y a + C0
(I)   e r + I0
(eq1) Y = e r + Y a + I0 + G0 + C0
(eq2) 250 = Y f - g r
(soln) [[Y =  $\frac{-I_0 g - G_0 g - C_0 g + 250 e}{a g - g + e f}$ , r =  $-\frac{I_0 f + G_0 f + C_0 f + 250 a - 250}{a g - g + e f}$ ]]
```

```
(%i173) soln : soln[1];
(soln) [Y =  $\frac{-I_0 g - G_0 g - C_0 g + 250 e}{a g - g + e f}$ , r =  $-\frac{I_0 f + G_0 f + C_0 f + 250 a - 250}{a g - g + e f}$ ]
```

Given the solutions for Y and r we can evaluate C and I .

```
(%i174) at (C, soln), ratsimp;
(%o174) -  $\frac{((I_0 + G_0) a + C_0) g - C_0 e f - 250 a e}{(a - 1) g + e f}$ 
```

```
(%i175) at(l, soln), ratsimp;
```

```
(%o175) 
$$\frac{(10a - 10)g + (-G0 - C0)ef + (250 - 250a)e}{(a - 1)g + ef}$$

```

8 Multiple Plots, Changing Plot Sizes with wxplot_size


Examples of using gr2d function from the draw package.

Examples of using wxdraw with gr2d to make multiple plots in the same cell.

Examples of using the external modifying option wxplot_size to get larger than the default size.

The default is one column of plots.

```
→ sinplot : gr2d( color = blue, key="sin (x)", grid=[2,2], explicit( sin(x), x, 0, 2*pi ) )$
cosplot : gr2d( color = red, key_pos = top_center, key="cos (x)",
  explicit(cos(x), x, 0, 2*pi ) )$
wxdraw ( sinplot, cosplot ), wxplot_size = [680, 680]$
```




```
(%t178)
```

Here we use the default size, and force two column display of the plots.

```
(%i179) wxdraw ( sinplot, cosplot, columns = 2);
```

```
(%t179)
```




```
(%o179)
```

Two column display with larger size.

```
(%i180) wxdraw ( sinplot, cosplot, columns = 2), wxplot_size = [880, 480];
```


```
(%t180)
```



```
(%o180)
```

default size, one column default (two rows).

```
(%i181) wxdraw ( sinplot, cosplot );
```



(%t181)


(%o181)

9 Proportional Axes Plots

Proportional Axes plots are useful if the line $y = x$ is to look like a 45 deg line. First we use `wxplot2d`, which has its own syntax (see Manual under `plot2d`). `plot2d` need the option `[same_xy, true]` for a proportional plot.

```
→ wxplot2d ([x, 4 + 0.2*x, 4 + 0.4*x, [ parametric, 5, t, [t, 0, 5]],
            [parametric, 20/3, t, [t, 0, 20/3]] ], [x, 0, 10], [style, [lines,2]],
            [same_xy, true], [grid2d, true], [ gnuplot_preamble, "set key bottom right;" ] )$
```


(%t182)



With `wxdraw2d` use `proportional_axes = xy` as an option.

```
(%i183) wxdraw2d (key_pos = top_left, proportional_axes = xy,  
  key = "y = x", explicit (x, x, 0, 10), color = red, key = "4 + 0.2*x",  
  explicit (4 + 0.2*x, x, 0, 10), color = green, key = "4 + 0.4*x",  
  explicit (4 + 0.4*x, x, 0, 10), color = magenta, key = "x = 5",  
  parametric (5, t, t, 0, 5), color = black, key = "x = 20/3",  
  parametric (20/3, t, t, 0, 20/3));
```


(%t183)



(%o183)

Increasing the size of the plot with wxdraw2d:

```
(%i184) wxdraw2d (key_pos = top_left, proportional_axes = xy,
  key = "y = x", explicit (x, x, 0, 10), color = red, key = "4 + 0.2*x",
  explicit (4 + 0.2*x, x, 0, 10), color = green, key = "4 + 0.4*x",
  explicit (4 + 0.4*x, x, 0, 10), color = magenta, key = "x = 5",
  parametric (5, t, t, 0, 5), color = black, key = "x = 20/3",
  parametric (20/3, t, t, 0, 20/3)), wxplot_size = [680, 680];
```




(%o184)

10 Colors in Plots

Default plot2d colors using wxplot2d. plot2d ([discrete, points], options), displays a list of points, joined by segments by default. The default colors start with blue, then red...

```
(%i185) wxplot2d ( [ [discrete, [ [0,0], [0,5] ] ], [discrete,[[2,0],[2,5]]],
    [discrete,[[4,0],[4,5]]], [discrete,[[6,0],[6,5]]], [discrete,[[8,0],[8,5]]], [discrete,[[10,0],[10,5]]],
    [discrete,[[12,0],[12,5]]], [discrete,[[14,0],[14,5]]] ], [x,-2,20], [y, -2,8],
    [style, [lines, 4] ] );
```

(%t185)




(%o185)

Here is an example of using the option [color, c1, c2, ...] with plot2d to customize the color rotation to be used.

```
→ wxplot2d ( [ [discrete, [ [0,0], [0,5] ] ], [discrete,[[2,0],[2,5]]],
    [discrete,[[4,0],[4,5]]], [discrete,[[6,0],[6,5]]], [discrete,[[8,0],[8,5]]],
    [discrete,[[10,0],[10,5]]], [discrete,[[12,0],[12,5]]], [discrete,[[14,0],[14,5]]] ],
    [x,-2,20], [y, -2,8], [style, [lines, 4] ],
    [color, blue, red, green, magenta, black, cyan, "#259452", "#25cd62"] );
```


(%t186)



(%o186)

The wxdraw2d default color is blue. Any other color choice must be explicitly given in the form `color = a_color_choice_understood_by_draw2d`. See the manual description of draw2d.

```
(%i187) wxdraw2d ( line_width = 5, yrange = [0,30], explicit (x,x,0,10),
  color = dark_magenta, explicit (x^1.5,x,0,10),
  color = "#259452", explicit (x^1.8, x, 0, 10) )$
```



11 Global & Local Variables with a Maxima Function

When using the block version of a Maxima function, the list `[a,b,c]` in our example lists the "local atomic symbols" which are private to the function and not accessible outside the function. Notice that the symbol `d` is not included in our list. `d` is defined inside the function and becomes a globally known function (known outside the function). The global symbol `abc` is defined before executing the function and that assignment is known and available inside the function.

```
(%i188) abc : 5;
```

```
(abc) 5
```

```
(%i190) kill(f,z)$
f(z) := block ([ a, b, c ],
  [a,b,c,d] : z*[1, 2, 3, 4]/abc,
  display (a,b,c,d, abc),
  done)$
```

```
(%i191)f(2)$
```

$$a = \frac{2}{5}$$

$$b = \frac{4}{5}$$

$$c = \frac{6}{5}$$

$$d = \frac{8}{5}$$

$$abc = 5$$

```
(%i192)print ("a =",a," b =",b," c =",c," d =",d," abc =", abc)$
```

$$a = a \quad b = b \quad c = c \quad d = \frac{8}{5} \quad abc = 5$$

```
(%i193)[a, b, c, d];
```

```
(%o193) [a,b,c,  $\frac{8}{5}$ ]
```